calculus_II

Αθροίσματα - Γινόμενα

```
n,k=var('n k')
```

```
sum(k^2,k,1,n)
```

$\frac{1}{3} n^3 + \frac{1}{2} n^2 + \frac{1}{6} n$

```
sum(1/k^2,k,1,infinity)
```

$\frac{1}{6} \pi^2$

```
sum((-1)^k/k,k,1,infinity)
```

-log(2)

```
sum(1/(n^2-1),n,3,infinity)
```

$\frac{5}{12}$

```
sum(1/k^4,k,1,infinity)
```

$\frac{1}{90} \pi^4$

```
sum(1/k^3,k,1,infinity)
```

$\zeta(3)$

```
zeta(3.)
```

1.20205690315959

Όρια

```
lim((2*x^2-x+1)/(3*x^2-5),x=0)
```

-1/5

```
lim((2*x^2-x+1)/(3*x^2-5),x=infinity)
```

2/3

```
lim(sin(x)/x,x=0)
```

1

```
lim(arctan(x),x=1)
```

1/4*pi

Ανάπτυγμα σε Σειρά

```
f=log(1+x);
f.taylor(x,0,10)
```

$-\frac{1}{10} x^{10} + \frac{1}{9} x^9 - \frac{1}{8} x^8 + \frac{1}{7} x^7 - \frac{1}{6} x^6 + \frac{1}{5} x^5 - \frac{1}{4} x^4 + \frac{1}{3} x^3 - \frac{1}{2} x^2 + x$

```
f1=arctan(x).series(x==0,10);f1
```

$1x + (-\frac{1}{3})x^3 + \frac{1}{5} x^5 + (-\frac{1}{7})x^7 + \frac{1}{9} x^9 + \mathcal{O}(x^{10})$

```
f2=sin(x).series(x==pi,4);f2
```

$(-1)(-\pi + x) + \frac{1}{6} (-\pi + x)^3 + \mathcal{O}\left((\pi - x)^4\right)$

```
f2.truncate()
```

$\pi - \frac{1}{6} (\pi - x)^3 - x$

```
diff(f1,x)
```

$1 + (-1)x^2 + 1x^4 + (-1)x^6 + 1x^8 + \mathcal{O}(x^9)$

```
expand(f1*f1)
```

$(1x + (-\frac{1}{3})x^3 + \frac{1}{5} x^5 + (-\frac{1}{7})x^7 + \frac{1}{9} x^9 + \mathcal{O}(x^{10}))^2$

Διανύσματα

```
a=vector(QQ,[1,2/3,-1]);
b=vector(QQ,[-3,1/3,2]);
c=vector(QQ,[-1,1,-1]);
```

```
a+b
```

(-2, 1, 1)

Εσωτερικό Γινόμενο

```
a.dot_product(b)
```

-43/9

Εξωτερικό γινόμενο

```
a.cross_product(b)
```

    (5/3, 1, 7/3)

```
a[0]
```

    1

```
b[1]
```

    1/3

```
norm(a)
```

    1/3*sqrt(22)

Κλίση, Απόκλιση, Περιστροφή

```
x,y,z=var('x y z');
f=(x-z)^2+y^2*x+z*x;
f.gradient([x,y,z])
```

$\left(y^{2} + 2\,x - z,\,2\,xy,\,-x + 2\,z\right)$

```
f=vector([-x^2*z, x,z*x])
divf=diff(f[0],x)+diff(f[1],y)+diff(f[2],z)
divf
```

$-2\,xz + x$

```
x,y=var('x y')
v=-x*sin(x+y)
ee=-v.gradient([x,y])
```

```
plot_vector_field(ee,(x,-2,2),(y,-2,2))
```

```
curl=vector([diff(f[2],y)-diff(f[1],z),diff(f[0],z)-diff(f[2],x), diff(f[1],x)-diff(f[0],y)]);
curl
```

$\left(0,\,-x^{2} - z,\,1\right)$

```
var ('x y z')
def divergence(F):
    assert(len(F) == 3)
    return diff(F[0],x) + diff(F[1],y) + diff(F[2],z)
def curl(F):
    assert(len(F) == 3)
    return vector([diff(F[2],y)-diff(F[1],z), diff(F[0],z)-diff(F[2],x),\
     diff(F[1],x)-diff(F[0],y)])
```

```
divergence(f)
```

$-2\,xz + x$

```
curl(f)
```

$\left(0,\,-x^{2} - z,\,1\right)$

Πίνακες

```
matrix.identity(3)
```

$$\left(\begin{array}{rrr} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{array}\right)$$

```
m=matrix(SR,3,3,[3,2,4,2,0,2,4,2,4]);
m
```

$$\left(\begin{array}{rrr} 3 & 2 & 4 \\ 2 & 0 & 2 \\ 4 & 2 & 4 \end{array}\right)$$

```
a=matrix(SR,3,3,[1,0,1,1,0,1,2,2,2]);a
```

$$\left(\begin{array}{rrr} 1 & 0 & 1 \\ 1 & 0 & 1 \\ 2 & 2 & 2 \end{array}\right)$$

```
a*m
```

$$\left(\begin{array}{rrr} 7 & 4 & 8 \\ 7 & 4 & 8 \\ 18 & 8 & 20 \end{array}\right)$$

```
m^(-1)
```

$$\left(\begin{array}{rrr} -1 & 0 & 1 \\ 0 & -1 & \frac{1}{2} \\ 1 & \frac{1}{2} & -1 \end{array}\right)$$

```
det(m)
```

4

```
m*m^(-1)
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
m.eigenvalues()
```

$\left[-2\sqrt{5} + 4, 2\sqrt{5} + 4, -1\right]$

```
m.eigenvalues()
```

$\left[-2\sqrt{5} + 4, 2\sqrt{5} + 4, -1\right]$

```
e=m.eigenvectors_right();e
```

$\left[\left(-2\sqrt{5} + 4, \left[\left(1, \frac{1}{2}, -\frac{1}{2}\sqrt{5}\right)\right], 1\right), \left(2\sqrt{5} + 4, \left[\left(1, \frac{1}{2}, \frac{1}{2}\sqrt{5}\right)\right], 1\right), \left(-1, \left[(1, -2, 0)\right], 1\right)\right]$

```
m.characteristic_polynomial()
```

$x^3 - 7x^2 - 12x - 4$

```
b=vector(SR,[1,2,-1]);m*b
```

$(3, 0, 4)$

```
e1=vector(e[0][1][0]);
e2=vector(e[1][1][0]);
e3=vector(e[2][1][0]);
```

```
expand(e1.dot_product(e2))
```

0

```
x=var('x');p=m.characteristic_polynomial();p
```

$x^3 - 7x^2 - 12x - 4$

Εύρεση πολυωνύμων Legendre από γεννήτρια συνάρτηση

```
t=var('t')
f(t)= 1/sqrt(1-2*x*t+t^2)
```

```
s=f(t).taylor(t,0,10).substitute(x=cos(t));s
```

$\frac{1}{256}\left(46189\cos(t)^{10} - 109395\cos(t)^{8} + 90090\cos(t)^{6} - 30030\cos(t)^{4} + 3465\cos(t)^{2} - 63\right)t^{10} + \frac{1}{128}\left(12155\cos(t)^{9} - 25740\cos(t)^{7} + 1801\right.$

```
p8=s.coefficient(t^8);
p3=s.coefficient(t^3)
```

```
plot(p3,(t,0,pi))+plot(p8,(t,0,pi),linestyle='--')
```



Ολοκλήρωση

```
integrate(sin(x)^2,x)
```

$\frac{1}{2}\,x - \frac{1}{4}\,\sin(2\,x)$

```
integrate(sin(x^2),x,0,Infinity)
```

$\frac{1}{4}\,\sqrt{2}\sqrt{\pi}$

```
a=var('a');
integrate(exp(-a*x^2),x,-Infinity,Infinity)
```

```
Traceback (click to the left of this block for traceback)
...
Is a positive, negative or zero?
```

```
assume(a>0);
integrate(exp(-a*x^2),x,-Infinity,Infinity)
```

$\frac{\sqrt{\pi}}{\sqrt{a}}$

```
y=var('y');
integrate(1/(x^2+y^2),x,0,1)
```

$\frac{\arctan\left(\frac{1}{y}\right)}{y}$

```
numerical_integral(exp(-x^2)*sin(x),0,1)
```

$\left(0.294698182249, 3.27180707248 \times 10^{-15}\right)$

```
numerical_integral(exp(-x^2)*sin(x),0,infinity)
```

$\left(0.424436383502, 5.8599714347 \times 10^{-08}\right)$

```
numerical_integral(exp(-x^2)*sin(x),0,1)[0]
```

0.294698182249

Διαφορικές Εξισώσεις : Αναλυτική Επίλυση

```
x = var('x'); y = function('y', x)
```

```
z=desolve(diff(y,x) + y == sin(x), y);z
```

$-\frac{1}{2}\,((\cos(x) - \sin(x))e^{x} - 2\,C)e^{(-x)}$

```
expand(z)
```

$Ce^{(-x)} - \frac{1}{2}\,\cos(x) + \frac{1}{2}\,\sin(x)$

```
yy=desolve(diff(y,x) + y+y^2 == 0, y);
yy
```

$\log(y(x) + 1) - \log(y(x)) = C + x$

```
solve(yy.simplify_log(),y(x))
```

$\left[y(x) = \frac{1}{e^{(C+x)} - 1}\right]$

```
g(x)=desolve(diff(y,x) + y == sinh(x), y,ics=[0,1]).expand();
g(x)
```

$-\frac{1}{2}\,xe^{(-x)} + \frac{3}{4}\,e^{(-x)} + \frac{1}{4}\,e^{x}$

```
g(0)
```

1

```
desolve(diff(y,x,2) + 4*diff(y,x)+3*y == sin(x), y,ics=[0,1,0])
```

$-\frac{1}{5}\,\cos(x) + \frac{7}{4}\,e^{(-x)} - \frac{11}{20}\,e^{(-3\,x)} + \frac{1}{10}\,\sin(x)$

```
desolve(diff(y,x,2) + 4*diff(y,x)+3*y == sin(x), y,ics=[0,1,1,0])
```

$\frac{(2\,\cos(1)e^{3} - e^{3}\sin(1) - 12)e^{(-x)}}{10\,(e^{2} - 1)} - \frac{(2\,\cos(1)e^{3} - e^{3}\sin(1) - 12\,e^{2})e^{(-3\,x)}}{10\,(e^{2} - 1)} - \frac{1}{5}\,\cos(x) + \frac{1}{10}\,\sin(x)$

Διαφορικές Εξισώσεις : Αριθμητική Επίλυση

Αριθμητική επίλυση συστήματος πρωτοβάθμιων διαφορικών εξισώσεων

Παρακάτω επιλύουμε αριθμητικά το σύστημα:

$$\frac{dx(t)}{dt} = x(t)^2 - y(t)$$

$$\frac{dy(t)}{dt} = x(t)y(t) + x(t)$$

με αρχικές συνθήκες $x(0) = 1, y(0) = 2$.

```
x,y = var('x y'); t=var('t');
tt=srange(0,30.05,0.05);
sol=desolve_odeint([x^2-y,x*y+x],[1,2],tt,[x,y],ivar=t)
```

```
g1=points(zip(tt,sol[:,0]),legend_label='$x(t)$');g1
```




```
g2=points(zip(tt,sol[:,1]),legend_label='$y(t)$',hue=0.9);g2
```




```
g1+g2
```




```
points(sol,axes_labels=['$x(t)$','$y(t)$'])
```



Αριθμητική επίλυση δευτεροβάθμιας διαφορική εξίσωσης

Παρακάτω επιλύουμε αριθμητικά την εξίσωση:

$$\frac{d^2x(t)}{dt^2} = -a\,x(t)^3 - b\,x(t) - \cos(\omega\,t)$$

με αρχικές συνθήκες $x(0) = 1, x'(0) = 0$.

Για το σκοπό αυτό μετατρέπουμε την δευτεροβάθμια εξίσωση σε σύστημα δύο εξισώσεων πρώτου βαθμού

$$\frac{dx(t)}{dt} = y(t)$$

$$\frac{dy(t)}{dt} = -a\,x(t)^3 - b\,x(t) - \cos(\omega\,t)$$

με αρχικές συνθήκες $x(0) = 1, y(0) = 0$

```
t=var('t');x=var('x');y=var('y');
a=2;b=1;w=1;
tt=srange(0,30.02,0.02);
sol=desolve_odeint([y,-a*x^3-b*x-cos(w*t)],[0,1],tt,[x,y],ivar=t)
```

```
points(zip(tt,sol[:,0]),legend_label='$x(t)$')+points(zip(tt,sol[:,1]),hue=0.4,legend_label='$x\'(t)$')
```




```
points(sol,axes_labels=['$x(t)$','$x\'(t)$'])
```





Σφαίρα μάζας m συνδέεται με ιδανικό άκαμπτο ελατήριο φυσικού μήκους ℓ και σταθεράς k.

Εκφράζοντας τους δύο βαθμούς ελευθερίας με τη γωνία θ και την επιμήκυνση x οι εξισώσεις κίνησης παίρνουν τη μορφή:

$$\ddot{x} = (\ell + x)\dot{\theta}^2 + g\,\cos\theta - \frac{k}{m}\,x$$

$$(l + x)\ddot{\theta} = -2\dot{x}\dot{\theta} - g\,\sin\theta$$

```
t=var('t');x=var('x');th=var('th');
dx=var('dx');dth=var('dth');
l=1;g=1;k=3;m=1;
tt=srange(0,30.02,0.02);
sol=desolve_odeint([dx,(l+x)*dth^2+g*cos(th)-(k/m)*x,dth,\
(-2*dx*dth-g*sin(th))/(l+x)],[0.4,0.1,9*pi/10,-0.3],tt,[x,dx,th,dth],ivar=t)
```

```
points(zip(tt,sol[:,0]),legend_label='$x(t)$')
```




```
yy=[-(l+sol[i][0])*cos(sol[i][2]) for i in srange(0,len(tt),1)];
xx=[(l+sol[i][0])*sin(sol[i][2]) for i in srange(0,len(tt),1)];
points(zip(xx,yy))
```




```
def pfun(ii):
    pd=[(xx[k],yy[k]) for k in srange(0,ii-1,8)]
    if ii>10:
        return line([(0,0),(xx[i],yy[i])],ymin=-4,ymax=2,xmin=-2,xmax=2)+\
circle((xx[i],yy[i]),0.07,fill=True)+line2d(pd,alpha=0.4,color='red')
    else:
        return line([(0,0),(xx[i],yy[i])],ymin=-4,ymax=2,xmin=-2,xmax=2)+
circle((xx[i],yy[i]),0.07,fill=True)
data=[pfun(i) for i in srange(1,len(tt),1)];
nof=25;
istep=int(len(data)/nof);
ddata=[data[i] for i in srange(0,len(data),istep)];
a=animate(ddata)
```

```
a.show()
```